
Abstract. This paper explores the use of Spin for the veri�cation of
cryptographic protocol security properties. A general method is proposed
to build a Promela model of the protocol and of the intruder capabilities.
The method is illustrated showing the modeling of a classical case study,
i.e. the Needham-Schroeder Public Key Authentication Protocol. Using
the model so built, Spin can �nd a known attack on the protocol, and it
correctly validates the �xed version of the protocol.

1 Introduction

All the solutions adopted to ensure security properties in distributed systems
are based on some kind of cryptographic protocol which, in turn, uses basic
cryptographic operations such as encryption and digital signatures. Despite their
apparent simplicity, such protocols have revealed themselves to be very error
prone, especially because of the diÆculty generally found in foreseeing all the
possible attacks. For this reason, researchers have been working on the use of
formal veri�cation techniques to analyze the vulnerability of such protocols.

Both the theorem proving and the model checking approaches have been
investigated. Some of the researchers who have investigated the model checking
approach have developed speci�c model checkers for cryptographic protocols (e.g.
[1]), whereas others have shown how general purpose tools such as FDR [2] and
Murphi [3] can be used for the same purpose. In this paper, we follow the latter
kind of approach, and we explore the possibility of using Spin, which is one of the
most powerful general purpose model checkers, to verify cryptographic protocols.
Instead of simply porting the approaches developed by other researchers to the
Spin environment, we develop a new approach which makes use of static analysis
techniques in order to get simpler models. The main idea is that the protocol
con�guration to be checked (i.e. the protocol sessions included in the model)
can be statically analyzed in order to collect data-
ow information, which can
be used to simplify the intruder knowledge representation. For example, such
a preliminary analysis can identify which data can potentially be learned by
the intruder and which cannot, thus avoiding the representation of knowledge
elements that will never occur. Similarly, it is possible to foresee which messages
that the intruder could build will never be accepted as valid by any protocol
agent, and avoid their generation.



The use of Spin for cryptographic protocol veri�cation has already been pro-
posed and discussed in [4], where, however, the author does not give a concrete
proposal, but just some general ideas and evaluations of the complexity of the
veri�cation task.

This paper is organized as follows. In section 2, we brie
y introduce the
Needham-Schroeder Public Key Authentication Protocol, which will be used
throughout the article to illustrate our modeling approach. In section 3, we
present the basic choices and the main underlying principles of our modeling
approach, whereas, in section 4, we give a detailed description of the procedure
to build a Promela model of an instance of a cryptographic protocol, using
the sample protocol as an example. In section 5, we present veri�cation results
related to the sample protocol. Section 6 concludes.

2 The Needham-Schroeder Public Key Protocol

The Needham-Schroeder Public Key Protocol [5] is a well known authentication
protocol that dates back to 1978. It aims to establish mutual authentication
between an initiator A and a responder B, after which some session involving
the exchange of messages between A and B can take place.

As its name clearly suggests, the protocol uses public key cryptography [6,
7]. Each agent H possesses a public key, denoted PK(H), and a secret key

SK(H), which can be used to decrypt the messages encrypted with PK(H).
While SK(H) should be known only by H , any other agent can obtain PK(H)
from a key server. Any agent can encrypt a message x using H 's public key
to produce the encrypted message fxgPK(H). Only the agents that know H 's
secret key can decrypt this message in order to obtain x. This property should
ensure x secrecy. At the same time, any agent H can sign a message x by
encrypting it with its own secret key, fxgSK(H), in order to ensure its integrity.
Any agent can decrypt fxgSK(H), using H 's public key.

The complete Needham-Schroeder Public Key Protocol [5] involves seven
steps and it is described in �gure 1, where A is an initiator agent who requests
to establish a session with a responder agent B and S is a trusted key server.

Any run of the protocol opens with A requesting B's public key to the trusted
key server S (step 1). S responds sending the message 2. This message, signed
by S to ensure its integrity, contains B's public key, PK(B), and B's identity.
If S is trusted, this should assure A that PK(B) is really B's public key. It is
worth noting that the protocol assumes that A can obtain PK(A), needed to
decrypt message 2, in a reliable way. If this assumption is not true, an intruder
could try to replace S providing an arbitrary value that A thinks to be PK(A).
Note also, that, as pointed out in [8], there is no guarantee that PK(B) is really
the current B's public key, rather that a replay of an old and compromised key
(however this attack can be easily prevented using timesptamps [8])

Fig. 1. The complete Needham-Schroeder Public Key Protocol.

Once obtained B's public key, A selects a nonce1 Na and sends message
3 to B. This message can only be understood by B, beeing encrypted with its
public key, and indicates that someone, supposed to be A, whishes to autheticate
himself to B.

After having received message 3, B decrypts it using its secret key to obtain
the nonce Na and then requests A's public key to S (steps 4 and 5).

At this point, B sends the nonce Na to A, along with a new nonce Nb,
encrypted with A'a public key (message 6). With this message, B authenticates
itself to A, since, receiving it, A is sure that it is communicating with B, being
B the only agent that should be able to obtain Na decrypting message 3.

To �nish the protocol run, A returns the nonce Nb to B in order to authen-
ticate itself to B (message 7).

Looking at the protocol, it is easy to observe as four of the seven steps can
be removed if we assume that A and B already know each other's public keys.
Indeed, messages 1, 2, 4 and 5 make up a protocol that aims to obtain A's and
B's public keys from a trusted key server S, whereas messages 3, 6 and 7 make
up the real authentication protocol.

In the following of this paper, we will assume that all the agents already know
each other's public keys and so we focus our attention on the reduced protocol
obtained removing messages 1, 2, 4 and 5 and described in �gure 2.

1 A nonce is a random number generated with the purpose to be used in a sinlge run
of the protocol.

Fig. 2. The reduced Needham-Schroeder Public Key Protocol.

3 Modeling cryptographic protocols with PROMELA

Formal models of cryptographic protocols are typically composed of a set of
principals which send messages to each other according to the protocol rules,
and an intruder, representing the activity of possible attackers.

Since such models are meant to reveal possible security 
aws in the proto-
cols and not 
aws in the cryptosystems used by the protocols, cryptography is
modeled in a very abstract way and it is assumed to be \perfect". This means
that:

{ the only way to decrypt an encrypted message is to know the corresponding
key;

{ an encrypted message does not reveal the key that was used to encrypt it;
{ there is suÆcient redundancy in messages so that the decryption algorithm
can detect whether a ciphertext was encrypted with the expected key.

Although such assumptions are obviously not completely true for real cryptosys-
tems, they represent the properties of an ideal cryptosystem, so they are useful
to isolate the 
aws of the protocol itself. In other words, any 
aw found with this
model is a real protocol 
aw, but it is possible that the model does not reveal
other weaknesses due to the used cryptosystems.

Both the principals and the intruder are represented in our models by means
of Promela processes that communicate with each other through shared channels.
More precisely, we have a di�erent process de�nition for each protocol role, and
a process de�nition for the intruder. Principals do not communicate with each
other directly but all the messages they send are intercepted by the intruder
which eventually will forward them to the right addressee. This approach has
been followed by other researchers too (e.g. [1]) and avoids redundant execution
paths.

The intruder can interact with the protocol in any way we would expect a
real-word attacker to be able to do, but at the same time it is also able to behave
like a normal user of the computer network. For this reason other principals may
initiate protocol runs with it. It is even possible that the intruder behaves as a
set of di�erent cooperating malicious users.

At any instant, the behavior of the intruder depends on the knowledge it has
acquired. Before a protocol run starts, it is assumed that the intruder knows
only a given set of data. For example, such data normally include the intruder



identity(ies), its public and private keys, the identity of the other principals,
their public keys, and, possibly, any secret keys the intruder shares with other
principals.

Every time the intruder intercepts a message, it can increase its knowledge.
Indeed, if the intercepted message or part of it is encrypted and the intruder
knows the decryption key, it can decrypt it and learn its contents. Otherwise, if
the intruder is not able to decrypt the intercepted message or parts of it, it can
remember the encrypted components even if it cannot understand them. Since
we are interested in modeling the most powerful intruder, we assume it always
learns as much as possible from the intercepted messages.

Of course, besides intercepting messages the intruder can also forge and send
new ones into the system. These ones are created using all the items it cur-
rently knows. Since it is normally assumed that the intruder can also create new
data items from scratch (e.g. nonces), and use them to forge new messages, this
capability can be represented including one or more generic distinguished data
items in the initial intruder knowledge. Such data items represent the ones the
intruder will generate from scratch during the protocol runs.

Note that, even if a message can be forged by the intruder, it could be that
it cannot be accepted by the receiving principal. This fact can be exploited to
safely restrict the messages the intruder can generate, excluding the ones that
will not be accepted by the receiving processes.

In order to have �nite models, we follow the common practice of putting
some restrictions on the modeled behaviors. First of all, the model can represent
only a �nite number of parallel protocol sessions. Each principal can be engaged
in a �nite number of runs of the protocol, each run being modeled by a di�erent
instance of the corresponding process de�nition. Nonces created during a proto-
col run are represented as process parameters, and are assigned di�erent actual
values at each process instantiation.

Another possible source of in�nite behaviors is the intruder. In fact, it can
in principle forge and send in�nite di�erent messages. A typical solution to this
problem is to restrict the way the intruder can generate messages, e.g. by restrict-
ing the maximum complexity of the generated messages [2, 1] and by limiting the
number of di�erent data items it can generate by scratch. More recent works [9]
have shown that such restrictions can be avoided by representing the messages
that can be generated by the intruder symbolically, and by specializing them as
soon as the receiving process performs some checks on them. In this paper we
follow a simpli�ed symbolic approach inspired by the one reported in [9], which
consists of symbolically representing all the data items the intruder can generate
from scratch as well as all the other complex data items it can generate from
its knowledge by means of a single distinguished symbolic identi�er. Such an
identi�er is always part of the intruder knowledge. However, instead of saying
that the intruder always sends a generic message which is then specialized, we
statically determine all the possible specializations it can take, which are �nite,
and we use the symbolic identi�er only for the leaves of the possible message
structures.



4 Building the Promela model

In this section we describe a procedure to build a Promela model of a crypto-
graphic protocol instance, to be used for security property veri�cation. Such a
procedure is illustrated using the reduced Needham-Schroeder Public Key Au-
thentication Protocol as an example. The Promela model we build can be divided
into two parts:

{ the description of the protocol rules and of the protocol instance
{ the description of the intruder behavior

The �rst part is quite simple and should be written manually, whereas the second
part can be generated automatically.

The instance of the sample protocol we deal with is fairly small. It includes
three principals: A, B and I . A plays the role of the initiator, whereas principal B
plays the role of the responder. I is one of the possible identities of the intruder,
so principal I can play any role.

4.1 The protocol instance model

The �rst step in the construction of the model is the de�nition of the �nite set
of names. With the term name we mean any distinguished identi�er, key, nonce
or data used in the protocol. As already explained, we also use a special name
which symbolically represents nonces and other generic data items generated by
the intruder. For our speci�c sample case, we need a name for each identity and
a name for each nonce. The resulting set of names we will use is then de�ned as
follows:

mtype = {A, B, I, Na, Nb, gD, R};

where A, B and I are the identities we consider, Na is the nonce generated by
principal A, Nb the nonce generated by B and gD the symbolic representation
of a generic data item used by the intruder. R is a service costant we will use in
the intruder process de�nition, as explained later on.

The second step in the construction of the model is the de�nition of the
channels used by the principals to communicate with each other. A di�erent
global channel is de�ned for each di�erent message structure used by the proto-
col. Of course, each protocol uses a �nite number of di�erent message structures,
which can be easily identi�ed by a simple inspection of the protocol messages.
For example, the reduced version of the Needham-Schroeder Public Key Au-
thentication Protocol uses only two message structures, i.e fx1; x2gPK(x3) (a
pair of elements encoded with a public key) and fx1gPK(x2) (a single element
encoded with a public key). Since we want to enforce the fact that messages
are always exchanged between the intruder and one of the other principals, we
always specify the identity of the principal involved in the communication with
the intruder as the �rst data exchanged on the channel, while we do not specify
the identi�er of the other party, because it is always the intruder. The subse-
quent data exchanged on the channel are the data components of the message



in order of occurrence in the message. So for example, the two global channels
for the sample protocol are de�ned as follows:

chan ca = [0] of {mtype, mtype, mtype, mtype};

chan cb = [0] of {mtype, mtype, mtype};

Channel ca is used to transfer messages of type fx1; x2gPK(x3), whereas channel
cb is used to transfer messages of type fx1gPK(x2). If, for example, the process
representing principal A has to send the message fNa;AgPK(B), the following
statement has to be used:

ca ! A, Na, A, B

where the �rst A represents the sender of the message, and the other three items
represent the message components (in this case, B is considered as a component
as well, since it uniquely identi�es PK(B)). The reception of a message by a
principal is expressed similarly. Here it is possible to express any requirements
on the received message as well. For example, to express the fact that A must
receive a message of type fx1; x2gPK(x3), and that it will accept as valid only
messages with x1 = Na and x3 = A, we can use the following statement:

ca ? eval(A), eval(Na), x2, eval(A);

where x2 is a local variable. In general, since all the messages sent by the intruder
have the indentity of the receiver as their left most item, when a normal host
has to receive a message, it must also require that the �rst data item is its name,
which means checking that the message is really addressed to it.

The next step in the model construction is the de�nition of the processes
representing the various roles of the protocol. Such processes must be parame-
terized with the data that may change from session to session and from instance
to instance. Their de�nition is the Promela representation of the sequence of
message exchanges prescribed by the protocol. The de�nition must include also
the recording of particular conditions that are useful in the expression of the
security properties to be checked. For our sample protocol, we de�ne proctype
PIni which describes the behavior of the initiator as follows:

proctype PIni (mtype self; mtype party; mtype nonce)

{

mtype g1;

atomic {

IniRunning(self,party);

ca ! self, nonce, self, party;

}

atomic {

ca ? eval (self), eval (nonce), g1, eval (self);

IniCommit(self,party);



cb ! self, g1, party;

}

}

Parameter self represents the identity of the host where the initiator process
is running, whereas party is the identity of the host with which the self host
wants to run a protocol session. Finally, nonce is the nonce that the initiator
process will use during the protocol run.

The atomic sequences have been used to reduce the amount of allowed in-
terleavings and so to reduce the complexity of the model from the veri�cation
point of view.

IniRunning and IniCommit are two macros used to update the values of the
variables recording the atomic predicates that are used to express the authenti-
cation properties. In order to explain this part, we have to explain the technique
used for property speci�cations, which is similar to the one presented in [2]. We
say that a protocol agent X takes part in a protocol run with agent Y if X has
initiated a protocol session with Y. Similarly, we say that a protocol agent X
commits to a session with agent Y if X has correctly concluded a protocol ses-
sion with Y. The fact that a responder with identity B correctly authenticates
to an initiator with identity A can be expressed by the following proposition:
A commits to a session with B only if B has indeed taken part in a run of the
protocol with A. A similar proposition expresses the reciprocal property, i.e. the
fact that an initiator with identity A correctly authenticates to a responder with
identity B.

Each one of the basic propositions involved in the above properties can be
represented in Promela by means of a global boolean variable which becomes
true at a particular stage of a protocol run. In the protocol con�guration we
want to analyze, we have to express that initiator A and responder B correctly
authenticate each other, so we need 4 variables, that we de�ne as follows:

bit IniRunningAB = 0;

bit IniCommitAB = 0;

bit ResRunningAB = 0;

bit ResCommitAB = 0;

IniRunningAB is true i� initiator A takes part in a session of the protocol with B.
ResRunningAB is true i� responder B takes part in a session of the protocol with
A. IniCommitAB is true i� initiator A commits to a session with B. ResCommitAB
is true i� responder B commits to a session with A.

Authentication of B to A can thus be expressed saying that ResRunningAB
must become true before IniCommitAB, whereas the converse authentication
property corresponds to saying that IniRunningABbecomes true before ResCommitAB.
In the LTL formalism, such precedence properties can be expressed as:

{ [] ( ([] !IniCommitAB) || (!IniCommitAB U ResRunningAB) )

{ [] ( ([] !ResCommitAB) || (!ResCommitAB U IniRunningAB) )



So, the macros IniRunning and IniCommit used in the protocol de�nition
update the values of the global variables IniRunningAB and IniCommitAB and
are de�ned in Promela as follows:

#define IniRunning(x,y) if \

:: ((x==A)&&(y==B))-> IniRunningAB=1 \

:: else skip \

fi

#define IniCommit(x,y) if \

:: ((x==A)&&(y==B))-> IniCommitAB=1 \

:: else skip \

fi

The proctype PRes of the responder processes is de�ned according to the
same principles as follows:

proctype PRes (mtype self; mtype nonce)

{

mtype g2, g3;

atomic {

ca ? eval (self), g2, g3, eval (self);

ResRunning(g3,self);

ca ! self, g2, nonce, g3;

}

atomic {

cb ? eval (self), eval (nonce), eval (self);

ResCommit(g3,self);

}

}

where parameter self represents the identity of the host where the responder
process is running and nonce is the nonce it will use during the protocol run.

ResRunning and ResCommit are macros used to update the values of the
global variables ResRunningAB and ResCommitAB and are de�ned as follows:

#define ResRunning(x,y) if \

:: ((x==A)&&(y==B))-> ResRunningAB=1 \

:: else skip \

fi

#define ResCommit(x,y) if \

:: ((x==A)&&(y==B))-> ResCommitAB=1 \

:: else skip \

fi



At this stage it is also possible to de�ne the protocol instance to be mod-
eled. It is simply speci�ed introducing a process instantiation statement in the
init process for each instance of the initiator and for each instance of the re-
sponder. The init process must include also the instantiation of a process PI
representing the intruder activity. PI is a process de�nition without parameters.
Its construction is illustrated in the following section.

For our sample instance, we have the following init de�nition:

init

{

atomic {

if

:: run PIni (A, I, Na)

:: run PIni (A, B, Na)

fi;

run PRes (B, Nb);

run PI ();

}

}

The if statement speci�es that principal A may initiate a protocol run with
any other principal, i.e. either with B or with I. A similar statement is not
needed for the responder, because it cannot decide the party with which it must
communicate.

4.2 The intruder model

The automatic construction of the intruder process de�nition requires a prelim-
inary static analysis in order to collect all the needed information.

First of all, we need to determine the sets of possible values taken by the
free variables occurring in each protocol process. For our sample protocol, such
variables are g1, g2 and g3. This operation can be performed using a simple
data-
ow analysis. For example, since initiator processes never check the value
of g1, but they simply pass it on, such an analysis will yield that in principle g1
can assume any possible value, i.e. Na, Nb, gD, A, B and I. The same argument
is valid for variable g2, while, since g3 is used as an host identi�er, the analysis
will yield that it can assume only the values A, B and I.

A second static analysis is then needed to restrict the potential intruder
knowledge representation to the minimum actually needed. First of all we have
to de�ne the intruder's initial knowledge, which, in our sample, is made up of the
identities and the public keys of all the principals of the system, i.e. A, B and I, the
intruder own private key, and the generic data gD. Such a knowledge can increase
when the intruder intercepts messages. If we compute all the possible messages
the intruder can intercept during the protocol run we can deduce also which are



Received message Learned item

fNa;AgPK(I) Na

fNa;AgPK(B) fNa;AgPK(B)

fNagPK(I) Na

fNbgPK(I) Nb

fgDgPK(I) -

fAgPK(I) -

fBgPK(I) -

fIgPK(I) -

fNagPK(B) fNagPK(B)

fNbgPK(B) fNbgPK(B)

fgDgPK(B) fgDgPK(B)

fAgPK(B) fAgPK(B)

fBgPK(B) fBgPK(B)

fIgPK(B) fIgPK(B)

fNa;NbgPK(I) Na, Nb

fNb;NbgPK(I) Nb

fgD;NbgPK(I) Nb

fNa;NbgPK(A) fNa;NbgPK(A)

fNb;NbgPK(A) fNb;NbgPK(A)

fgD;NbgPK(A) fgD;NbgPK(A)

fA;NbgPK(I) Nb

fB;NbgPK(I) Nb

fI;NbgPK(I) Nb

fA;NbgPK(A) fA;NbgPK(A)

fB;NbgPK(A) fB;NbgPK(A)

fI;NbgPK(A) fI;NbgPK(A)

Table 1. Knowledge elements that the intruder can eventually acquire

the possible messages the intruder can add to its knowledge. For example, if the
intruder intercepts the message fNa;AgPK(I) it can learn the nonce Na. In
order to avoid the representation of redundant knowledge elements, we assume
that the intruder always records the learned items in their most elementary
forms. For example, if message fNa;AgPK(I) is intercepted, we assume that
the intruder records only Na, and not the whole message fNa;AgPK(I), since
this message can be built from Na and from A, which is always known to the
intruder. In other words, the intruder records a complex message in its knowledge
only if it cannot decrypt it. For example, if the intruder intercepts the message
fNa;AgPK(B), it can only remember the message as a whole.

Knowledge items are represented in the intruder process PI by means of
local boolean variables. For atomic data items (names) such variables have the
same name of the data item itself, with the 'k' pre�x. For example, variable
kNa represents the knowledge of Na. Similarly, the knowledge of structured data
items can be represented by means of bit variables whose names have the 'k'



Message Needed knowledge (besides initial knowledge)

fNa;AgPK(B) Na or fNa;AgPK(B)

fNa;BgPK(B) Na or fNa;BgPK(B)

fNa; IgPK(B) Na or fNa; IgPK(B)

fNb;AgPK(B) Nb or fNb; AgPK(B)

fNb;BgPK(B) Nb or fNb; BgPK(B)

fNb; IgPK(B) Nb or fNb; IgPK(B)

fgD;AgPK(B) -

fgD;BgPK(B) -

fgD; IgPK(B) -

fNa;AgPK(A) Na or fNa;AgPK(A)

fNa;BgPK(A) Na or fNa;BgPK(A)

fNa; IgPK(A) Na or fNa; IgPK(A)

fA;AgPK(B) -

fA;BgPK(B) -

fA; IgPK(B) -

fB;AgPK(B) -

fB;BgPK(B) -

fB; IgPK(B) -

fI;AgPK(B) -

fI;BgPK(B) -

fI; IgPK(B) -

fNbgPK(B) Nb or fNbgPK(B)

fNa;NagPK(A) Na or fNa;NagPK(A)

fNa;NbgPK(A) (Na and Nb) or fNa;NbgPK(A)

fNa; gDgPK(A) Na or fNa; gDgPK(A)

Table 2. Knowledge elements potentially needed by the intruder

pre�x and include the various data �elds in order of occurrence. For example,
variable k Na A B will represent the knowledge of message fNa;AgPK(B).

Since the messages the intruder can intercept are �nite, we will also have a
�nite number of corresponding possible knowledge elements. Table 1 lists all the
possible messages that the intruder can intercept in our sample protocol instance
and, for each of them, the data items the intruder can learn from it.

A further restriction of the intruder knowledge items to be recorded is pos-
sible if we exclude the ones which can never be used by the intruder to generate
valid messages. For example, if the intruder knows message fNagPK(B) as a
whole, it can potentially send it to B, but this one will not accept it, so it is not
useful for the intruder to know that message. The set of data items potentially
useful to the intruder can be computed listing all the valid messages that the
intruder could eventually send to the other principals and determining, for each
of them, which data items the intruder could use to build it. The result of this
analysis is showed in Table 2 for our sample protocol instance.



The actual set of knowledge elements to be stored in the intruder is the
intersection of the two sets just computed (the right columns of Tables 1 and 2):

{ Nonces: Na and Nb;
{ Messages as a whole: fNa;NbgPK(A), fNa;AgPK(B) and fNbgPK(B).

Preformed these preliminary analyses, the PI proctype de�nition can be writ-
ten as follows:

proctype PI ()

{

/* The intruder always knows

A, B, I, PK(A), PK(B), PK(I), SK(I) and gD

*/

bit kNa = 0; /* Intruder knows Na */

bit kNb = 0; /* Intruder knows Nb */

bit k_Na_Nb__A = 0; /* " " {Na, Nb}{PK(A)} */

bit k_Na_A__B = 0; /* " " {Na, A}{PK(B)} */

bit k_Nb__B = 0; /* " " {Nb}{PK(B)} */

mtype x1 = 0, x2 = 0, x3 = 0;

do

:: ca ! B, gD, A, B

:: ca ! B, gD, B, B

:: ca ! B, gD, I, B

:: ca ! B, A, A, B

:: ca ! B, A, B, B

:: ca ! B, A, I, B

:: ca ! B, B, A, B

:: ca ! B, B, B, B

:: ca ! B, B, I, B

:: ca ! B, I, A, B

:: ca ! B, I, B, B

:: ca ! B, I, I, B

:: ca ! (kNa -> A : R), Na, Na, A

:: ca ! (((kNa && kNb) || k_Na_Nb__A) -> A : R),

Na, Nb, A

:: ca ! (kNa -> A : R), Na, gD, A

:: ca ! (kNa -> A : R), Na, A, A

:: ca ! (kNa -> A : R), Na, B, A

:: ca ! (kNa -> A : R), Na, I, A

:: ca ! ((kNa || kNa_A__B) -> B : R), Na, A, B

:: ca ! (kNa -> B : R), Na, B, B

:: ca ! (kNa -> B : R), Na, I, B

:: ca ! (kNb -> B : R), Nb, A, B



:: ca ! (kNb -> B : R), Nb, B, B

:: ca ! (kNb -> B : R), Nb, I, B

:: cb ! ((k_Nb__B || k_Nb) -> B : R), Nb, B

:: d_step {

ca ? _, x1, x2, x3; if

:: (x3 == I)-> k(x1);

k(x2)

:: else k3(x1,x2,x3)

fi;

x1 = 0;

x2 = 0;

x3 = 0;

}

:: d_step {

cb ? _, x1, x2; if

:: (x2 == I)-> k(x1)

:: else k2(x1,x2)

fi;

x1 = 0;

x2 = 0;

}

od

}

Let us �rst comment the variable declaration part. We have a bit variable
for each knowledge item to be represented in the intruder. These variables are
initially set to 0, because initially the corresponding data items are not known
to the intruder. They will be set to 1 as soon as the intruder learns the cor-
responding item. Besides the variables used to represent the knowledge of the
intruder (the ones whose name begins with k), three services variables x1, x2
and x3 are declared. The use of these ones will be explained later on.

Let us now consider the behavior description. The intruder behaves as a never
ending process that spends all its time sending messages to and receiving mes-
sages from the protocol channels (ca and cb). Since each operation is atomic, the
state of the intruder is determined only by the current contents of its knowledge
variables (service variables are always reset to 0 after each use).

Each branch of the main repetition construct represents an input or output
operation on the global channels (ca and cb).

There is one output branch for each possible message sent by the intruder.
Some output branches have a pre-condition which enables them, while other
output branches are not conditioned by the intruder knowledge, because the
intruder always knows how to build the corresponding messages. For example,
the branch:

:: ca ! B, gD, A, B



representing the intruder sending the message fgD;AgPK(B) to B, is always
executable if a process, in this case the process associated to the principal B, is
ready to synchronize with the intruder process. In contrast, the branch:

:: ca ! (kNa-> A : R), Na, Na, A

representing the intruder sending the message fNa;NagPK(A) to A, requires
also that the intruder knows Na, i.e. that the value of the local variable kNa

is 1. The conditional expression in the �rst position evaluates to A only if the
intruder knows Na, and evaluates to a the non-existing identity R otherwise.
This correctly means that the intruder can send message fNa;NagPK(A) to A
i� it knows Na.

For what concerns input branches, there is one of them for each channel. Each
input branch includes the input operation, which records the message compo-
nents in the service variables, and a series of subsequent decoding operations,
which depends directly on the message structure. The setting of knowledge vari-
ables is technically obtained by the following macros, which automatically ignore
useless knowledge elements:

#define k(x1) if \

:: (x1 == Na)-> kNa = 1 \

:: (x1 == Nb)-> kNb = 1 \

:: else skip \

fi;

#define k2(x1,x2) if \

:: (x1 == Nb && x2 == B)-> k_Nb__B = 1 \

:: else skip \

fi

#define k3(x1,x2,x3) if \

:: (x1 == Na && x2 == A && x3 == B) \

-> k_Na_A__B = 1 \

:: (x1 == Na && x2 == Nb && x3 == A) \

-> k_Na_Nb__A = 1 \

:: else skip \

fi

So, for example, if the intruder receives the message fNa;AgPK(I) from the
channel ca, the variable kNa is set to 1, and, if it receives the message fNbgPK(B)
from channel cb, the variable k Nb B is set to 1.

Note that input operations are included in d step sequences and that vari-
ables x1, x2 and x3 are always set to 0 before the end of the d step sequences.
This has been done to reduce the amount of possible states of the intruder pro-
cess.



Fig. 3. The Lowe's attack on the Needham-Schroeder Public Key Authentication Pro-
tocol found by Spin.

5 Veri�cation results

Analyzing the previously described model with Spin (we used version 3.4.10),
we were able to discover the attack described in [2] by Lowe. The veri�cation
of each property took a fraction of second. The �rst error trail showing the
attack is reported in �gure 3. As expected, verifying the model of the �xed
version of the protocol proposed in [2], we did not �nd any additional attack on
the protocol. The number of reachable states and transitions of the two models
(original and �xed protocol) for the veri�cation of the �rst property (A correctly
authenticates to B) are reported in table 3. In both cases, the models are referred
to a con�guration with one initiator, one responder and one intruder.

To evaluate the eÆciency of our modeling approach, we have compared our
results with the ones obtained analyzing the same instance of the Needham-
Schroeder protocol using the model checker Murhi [3] and the CSP base tools
Casper [10] and FDR [11] as described in [12]. All the approaches are able to
discover the Lowe's attack but there are some di�erences in the complexity of
the obtained models. The number of reachable states of our model is lower than
the Murphi's one, while it is higher than the one obtained using FDR. The lower
number of states of the FDR model is probably due to the fact that Casper



Modeled protocol States Transitions

Original version (property 1) 381 1195

Fixed version (property 1) 378 1275

Table 3. Veri�cation results

forces strict data typing, and then does not consider possible attacks due to
type confusion, which instead are considered by our modeling approach.

6 Conclusions

We have presented a way to model cryptographic protocols using Promela. The
modeling approach we propose consists of specifying the protocol rules and the
con�guration to be checked directly in Promela. Instead, the model of the in-
truder, which is the most diÆcult part, can be constructed automatically. We
described a procedure for the automatic generation of the intruder de�nition.
Such a procedure uses complexity reduction techniques based on a preliminary
data-
ow analysis to build a simpli�ed model. Following this approach we suc-
ceeded in �nding the well-known attack on the Needham-Shroeder Public Key
Authentication Protocol.

In this paper we have informally illustrated the procedure to construct the
intruder model, using a case study. Future work includes a formal de�nition of the
intruder construction procedure and its implementation in an automatic intruder
model generator. Another possible future development is the construction of
a user-friendly speci�cation interface which makes it possible to describe the
protocol rules and con�guration more directly and generate the whole Promela
model automatically.

